Warlord Mission Tutorial

To complete this tutorial you will modify the Stratis Island map and modify a mission script.

Preparing to Create the Mission

Copy the MissionWarlord.Stratis subdirectory to:

<MyDocuments>\Arma 3 - Other Profiles\<MyProfile>\missions\

Copy the Zen_FrameworkFunctions directory from the Shell.Stratis directory to:

<MyDocuments>\Arma 3 - Other Profiles\<MyProfile>\missions\ MissionWarlord.Stratis\

This directory (and its sub-directories) contain all the code for the Co-op Framework.

This is what your \missions\ MissionWarlord.Stratis directory structure should look like:



These are the contents of the \MissionWarlord.Stratis directory:



Updating the Map

Open Arma and then the Editor. Select Stratis Island and Continue.

Choose the Load function and select MissionWarlord.

To the Southeast of Girna is a two man BLUFOR unit. This is the squad you will command when you play this mission. This tutorial will discuss saving and starting this mission as a co-op mission.

If necessary switch to tradition view by selecting 

Select ADVANCED INTEL from sidebar menu:

Enter a time of 2145 hours. This is a night mission. The game will automatically outfit all members of the squad with the correct equipment: night vision goggle, laser sights, flashlights, etc.

Use F6 to enter the editor's Marker Mode. Double click in Girna (NW of squad).



The area marker is a circle.

Approximate relation of area marker to player team:

Choose Save or CTRL-S to save the mission.

You're finished with the updates that you will make to the map. But leave the editor open.

Updating the Initialization Script

Open the *init.sqf* file.

Immediately above the lines

// All clients stop executing here, do not delete this line
if (!isServer) exitWith {};

Add a mission briefing statement:

```
Player creatediaryRecord["Diary", ["Warlord Tutorial", "An evil
warlord must be killed.<br/>A handful of squads protect the warlord.
One patrols nearby, the others within a range of 100 to 250
meters.<br/>"]];
```

At the point labeled ‘Enter the mission code here’ enter these statements.

Generate a random point within the WarlordMissionOpfor area marker:

```
_ObjectivePos = ["WarlordMissionOpfor"] call Zen_FindGroundPosition;
```

Place a Warlord task objective at this random point.

```
_yourObjective = [_ObjectivePos, (group X11), east,
"Officer","eliminate"] call Zen_CreateObjective;
```

Place a guard on patrol near the Warlord and track:

```
_InnerPos = [_ObjectivePos, [70,150]] call Zen_FindGroundPosition;
_InnerGuard = [_InnerPos, east, "infantry", [2,3]] call
Zen_SpawnInfantry;
0 = [[_InnerGuard], _InnerPos, [70,150]] spawn
Zen_OrderInfantryPatrol;
_returnArray = [[_InnerGuard], "group"] call Zen_TrackInfantry;
```

Create squads to patrol further out:

```
_OuterPos = [_ObjectivePos, [120,170]] call Zen_FindGroundPosition;
_OuterGuard = [_OuterPos, east, "infantry", [2,3]] call
Zen_SpawnInfantry;
0 = [[_OuterGuard], _ObjectivePos, [120,170]] spawn
Zen_OrderInfantryPatrol;
_returnArray = [[_OuterGuard], "group"] call Zen_TrackInfantry;

_OuterPos = [_ObjectivePos, [170,250]] call Zen_FindGroundPosition;
_OuterGuard = [_OuterPos, east, "infantry", [2,3]] call
Zen_SpawnInfantry;
0 = [[_OuterGuard], _ObjectivePos, [170,250]] spawn
Zen_OrderInfantryPatrol;
_returnArray = [[_OuterGuard], "group"] call Zen_TrackInfantry;

_OuterPos = [_ObjectivePos, [170,250]] call Zen_FindGroundPosition;
_OuterGuard = [_OuterPos, east, "infantry", [2,3]] call
Zen_SpawnInfantry;
```

```
0 = [[_OuterGuard], _ObjectivePos, [170,250]] spawn
Zen_OrderInfantryPatrol;
_returnArray = [[_OuterGuard], "group"] call Zen_TrackInfantry;
```

Wait until the objective is complete and then end the mission:

```
waituntil { sleep 5; [(_yourObjective select 1)] call
Zen_AreTasksComplete };
endMission "end1"
```

Save the init.sqf file and return to the Arma editor.

Play the Mission.

To play this mission solo launch the mission from inside the editor by selection ‘Preview’.

Post-Mortem

If you played the mission here’s what you should have seen:

- A briefing
- A task for the “Kill the Warlord” objective.
- The marker area shown as a grey circle.
- A dot that showed approximately the location of the warlord.
- The patrolling squads are shown marked on map.
- After killing the warlord, the task should have shown completed.

Technical Corner

This tutorial shows the use of a specified start time, the creation of squads and how to track them.

Spawn Infantry and Patrol Order

First, for each OPFOR squad, a position is randomly generated within a range of distances from a single point. In this mission the single point (the objective position) is the location of the warlord, which is referenced by the variable “_ObjectivePos”. Then each OPFOR squad is spawned at that randomly generated position.

The spawned squad is then ordered to patrol within a range around the objective position. In this mission, the range of distances for the spawning points and for the patrol area are the same, but they can be different.

This donut shaped patrol area does not strictly constrain the squad; the squad can move out of the ‘donut’. But the Framework attempts to generate random waypoints that keep the squad within the donut. Enemy squad AI naturally overrides this. When the enemy AI spot a valid target (usually you!), they will seek and attack until their target is lost. Then they will move to the last waypoint generated by the Framework.

Avoid ordering patrols to patrol in areas with many fences and steep cliffs. They can become 'stuck' and will not advance to their waypoint.

Mass Spawing

Multiple squads can be spawned in the same area with a for-loop:

```
for [{_i=0}, {_i<4}, {_i=_i+1}] do {
      _OuterPos = [_ObjectivePos, [170,250]] call
      Zen_FindGroundPosition;
      _OuterGuard = [_OuterPos, EAST, "infantry", [2,3]] call
      Zen_SpawnInfantry;
      0 = [[_OuterGuard], true, _ObjectivePos, [180,230]] spawn
      Zen_OrderInfantryPatrol;
};
```

Combined Patrol Orders

Multiple squads can be combined into an array with the sqf 'set' command and then given patrol orders with a single call to Zen_OrderInfantryPatrol:

```
for "_i" from 0 to 1 do {
      _innerPos = [_objectivePos, [90,160]] call
      Zen_FindGroundPosition;
      _innerGuard = [_innerPos, east, "infantry", [2,3]] call
      Zen_SpawnInfantry;
      _innerGuardsGroupArray set [(count _innerGuardsGroupArray),
      _InnerGuard];
};

0 = [_innerGuardsGroupArray, _objectivePos, [70,180]] spawn
Zen_OrderInfantryPatrol;
```

Tracking Squads and Tracking Self

Multiple squads that are combined in an array can be tracked with one call to *Zen_TrackInfantry*:

```
_returnArray = [(_innerGuardsGroupArray + _outerGuardsGroupArray),
"group"] call Zen_TrackInfantry;
```

If you want to play on Veteran or Elite and track your own movements add this line:

```
_returnArray = [[(group X11)], "name"] call Zen_TrackGroups;
```